Paper SAS4626-2020

Introduction to SAS/ACCESS® Interface to Google BigQuery Joel Odom, SAS Institute Inc.

ABSTRACT

Google BigQuery is a service running on the Google Cloud Platform that facilitates analysis of massive data sets working in tandem with Google Cloud Storage. SAS/ACCESS® Interface to Google BigQuery enables SAS® to take advantage of this exciting technology. This paper describes Google BigQuery and discusses how it is different from other databases that you might have used in the past. Using examples, we discuss the following topics: how to get started connecting your SAS software to Google BigQuery; tricks to get the most out of SAS/ACCESS Interface to Google BigQuery; how to effectively move data into and out of Google BigQuery; and potential gotchas (or idiosyncrasies) of Google BigQuery. This paper uses an example-driven approach to explore these topics. Using the examples provided, you can apply what you learn from this paper to your environment.

INTRODUCTION

SAS/ACCESS is set of products through which the SAS system accesses DBMS data. SAS/ACCESS Interface to Google BigQuery enables you to connect to a BigQuery instance running in the Google Cloud Platform with minimal configuration steps and no additional software. This paper guides you through the configuration steps to connect to Google BigQuery and then demonstrates how to navigate your data and work with it natively from SAS.

CONFIGURATION

There is very little configuration required for connecting to Google BigQuery with SAS/ACCESS. The only requirement is that the BigQuery service be turned on in your GCP project. By default, per https://cloud.google.com/service-usage/docs/enabled-service, this should be turned on. However, it can be turned off, so if you have issues connecting, it's worth double checking that the service is turned on.

In the first release of SAS/ACCESS to Google BigQuery, on SAS Viya 3.4 and SAS 9.4M6, the only authentication method available is using a service account. To connect using this method, the credentials file associated with the service account that you are connecting with is required. This is obtained when creating a key on a service account. In an upcoming update to SAS/ACCESS to Google BigQuery, on SAS Viya 3.5 and SAS 9.4M6, support for OAuth authentication is being added. Support for OAuth authentication removes the need for the credentials file and provides an alternative authentication method.

CONNECTING

There are two required options for SAS/ACCESS Interface to Google BigQuery:

• PROJECT = – This option specifies the project that you want to use to connect to Google BigQuery. This project is used to charge queries against as well as other charges required by Google BigQuery. If the project name contains spaces or non-alphanumeric characters, you must enclose it in quotation marks. It is easier if the data you want to access is stored within this project, but this is not required. You can use the

TABLE_QUALIFIER= option to specify another project that you want to access data in.

CREDFILE= (alias: CRED_FILE=, CREDPATH=, CRED_PATH=) – This option specifies the path (including the filename) of the credentials file that is used to connect to Google BigQuery. This option can be left off if you set the environment variable GOOGLE_APPLICATION_CREDENTIALS to the path of the credentials file instead. As mentioned earlier, an upcoming update to SAS/ACCESS to Google BigQuery on SAS Viya 3.5 and SAS 9.4M6 adds support for OAuth authentication, which removes the need of the credentials file and makes CREDFILE= optional.

In addition to these two options, it might also be useful to set the following option:

SCHEMA= - This option specifies the Google BigQuery data set to use when loading and accessing tables in Google BigQuery. Like the PROJECT= option, if the dataset name contains spaces or non-alphanumeric characters, you must enclose it in quotation marks. Google BigQuery does not have default data sets, so this option allows you to specify a default to use per connection. If you do not specify this option on your connection or at code execution time, then you see the following error:

```
ERROR: File libref>..DATA does not exist.
```

LIBNAME STATEMENT

The SAS/ACCESS Interface to Google BigQuery LIBNAME statement extends the SAS global LIBNAME statement so that a libref can be assigned to Google BigQuery. This feature lets you reference a Google BigQuery table or Google BigQuery dataset directly in a DATA step or SAS procedure. You can use the defined libref to read from and write to Google BigQuery as if it were a SAS data set.

```
libname mydb bigquery project="sas-example"
credfile="/home/user/bigquery/sas-example-3e48c10a5978.json"
schema="bigquery_dataset";
```

In this example, the SCHEMA= option is specified so that the connection has a default Google BigQuery dataset. This means that you don't have to specify the Google BigQuery dataset at code execution time.

CASLIB STATEMENT

The CASLIB statement enables you to access Google BigQuery directly from the CAS server in SAS Viya. The CASLIB statement is very similar to the LIBNAME statement. One major difference is that the CASLIB statement does not attempt to create a connection when a statement is executed. The connection is deferred until the first request to access data in Google BigQuery is made by a CAS action. Specifying data connector options is done within the DATASOURCE= option. Using the above Google BigQuery LIBNAME statement, here is how to create a similar CASLIB statement:

```
caslib caslibBigQuery sessref=mysess
datasource=(srctype="bigquery",
project="sas-example",
credfile="/home/user/bigquery/sas-example-3e48c10a5978.json",
schema="bigquery_dataset");
```

The caslib (caslibBigQuery) name that is specified in the CASLIB statement must be unique within the session that is specified in the SESSREF option. If the caslib name that is specified is the same as a global caslib name, then the session caslib supersedes the global caslib, effectively making data in the global caslib inaccessible from within the session specified in the SESSREF option.

PROC FEDSOL AND PROC DS2

Accessing data in Google BigQuery through either the FEDSQL or DS2 procedure can be accomplished using one of two methods. The typical method is to use a LIBNAME statement and then reference the libref in the reference to the table that you want to access in Google BigQuery. The less common method uses the CONN= option in conjunction with the NOLIBS option.

By default, PROC FEDSQL and PROC DS2 ather all available librefs and attempt to make connections to the various data sources defined by those librefs. The NOLIBS option tells PROC FEDSQL and PROC DS2 to alter the default behavior and to ignore any librefs that are available in the current SAS session. The CONN= option then allows the user to specify which data sources they would like to connect to and with which options by specifying a connection string.

The second method allows the user to make a connection that is scoped only to the PROC FEDSQL or PROC DS2 job. This is very similar to the CONNECT TO option in PROC SQL. If you specify the NOLIBS option without the CONN= option, then the PROC FEDSQL or PROC DS2 job fails. The following examples make similar connections to Google BigQuery as the LIBNAME and CASLIB statement sections above using the second method:

```
proc fedsq1 nolibs conn='DRIVER=BIGQUERY;
PROJECT=sas-example;
CREDFILE=/home/user/bigquery/sas-example-3e48c10a5978.json;
SCHEMA=bigquery_dataset;
CATALOG=(mydb=sas-example)';
quit;

proc ds2 nolibs conn='DRIVER=BIGQUERY;
PROJECT=sas-example;
CREDFILE=/home/user/bigquery/sas-example-3e48c10a5978.json;
SCHEMA=bigquery_dataset;
CATALOG=(mydb=sas-example)';
quit;
```

In the connection string used with the CONN= option, there are two additional options, DRIVER= and CATALOG=. The DRIVER= option specifies which SAS data connector to use with PROC FEDSQL or PROC DS2, similar to the SRCTYPE= option in the DATASOURCE= option of the CASLIB statement. The CATALOG= option specifies an identifier for an internal SQL catalog to group logically related schemas. Since Google BigQuery has a concept that logically groups related schemas, called projects, we mapped the CATALOG= option to Google BigQuery projects. Also, a connection to Google BigQuery can access multiple projects, so the SAS data connector for Google BigQuery supports multi-catalog connections. To facilitate this, specify a list of catalogs in the connection string, similar to the following examples:

```
proc fedsql nolibs conn='DRIVER=BIGQUERY;
PROJECT=sas-example;
CREDFILE=/home/user/bigquery/sas-example-3e48c10a5978.json;
SCHEMA=bigquery_dataset;
CATALOG=(mydb=sas-example;mydb2=example2;"bigquery-public-data"={bigquery-public-data})';
quit;

proc ds2 nolibs conn='DRIVER=BIGQUERY;
PROJECT=sas-example;
CREDFILE=/home/user/bigquery/sas-example-3e48c10a5978.json;
SCHEMA=bigquery_dataset;
CATALOG=(mydb=sas-example;mydb2=example2;"bigquery-public-data"={bigquery-public-data})';
```

```
quit;
```

Another method to specify multiple catalogs is to allow a catalog map to be set such that all of the "native catalogs" (projects) are available to PROC FEDSQL or PROC DS2. To do this, set the CATALOG= option to "*", as shown in the following examples:

```
proc fedsql nolibs conn='DRIVER=BIGQUERY;
PROJECT=sas-example;
CREDFILE=/home/user/bigquery/sas-example-3e48c10a5978.json;
SCHEMA=bigquery_dataset;
CATALOG=*';
quit;

proc ds2 nolibs conn='DRIVER=BIGQUERY;
PROJECT=sas-example;
CREDFILE=/home/user/bigquery/sas-example-3e48c10a5978.json;
SCHEMA=bigquery_dataset;
CATALOG=*';
quit;
```

READING DATA

READING DATA WITHIN SAS 9

Using SAS/ACCESS Interface to Google BigQuery to read data is similar to any other SAS/ACCESS product.

DATA Step

One of the benefits of using SAS to access data in a third-party data source is that it handles most of the complexity that comes with accessing data in each data source. This is evident when using the SAS DATA step with SAS/ACCESS Interface to Google BigQuery. A lot can be accomplished with a simple four-line SAS program, such as the one below which creates a Stocks SAS data set and moves all stock data for Google using the LIBNAME defined previously:

```
data work.Google;
    set mydb.stocks(keep=stock date close);
    where stock = 'GOOGL';
run;
```

PROC SQL

The standard way to query a data source in SAS with SQL is through the SQL procedure. Here is an example of how to accomplish the same task shown previously using PROC SQL:

```
proc sql;
    create table work.Google as
        select stock, date, close
        from mydb.stocks
        where stock='GOOGL';
quit;
```

Explicit SQL Pass-Through

In PROC SQL, to query a data source using data source specific SQL that you specify, use the SQL pass-through facility. There are two methods that enable you to use the SQL pass-through facility: the CONNECT USING statement (recommended) and the CONNECT TO statement.

CONNECT USING

The following example uses SQL code that you provide to the SQL pass-through facility through the CONNECT USING statement with the LIBNAME defined previously:

```
proc sq1;
    CONNECT USING mydb;
    create table work.sasbigqueryclass as select * from
        connection to mydb
            (select * from bigquery_dataset.bigqueryclass);
    execute (drop table bigquery_dataset.bigqueryclass) by mydb;
    disconnect from mydb;
quit;
```

In this example, the proc SQL job executes two statements. The first creates the Work.Sasbigqueryclass table and loads the Google BigQuery data from the Bigqueryclass table into it. Unlike typical PROC SQL processing, the SELECT statement in the parenthesis is passed directly to Google BigQuery without SAS processing any part of it. The second statement drops the Bigqueryclass table in Google BigQuery. As the SQL in both statements above is passed directly to Google BigQuery as is, it must be valid Google BigQuery SQL.

One advantage to the CONNECT USING statement is that the user does not need to have any knowledge of the connection options used in the LIBNAME statement because it determines which data source to access through a libref.

CONNECT TO

The CONNECT TO statement is similar to connecting with PROC FEDSQL using the NOLIB and CONN= options described earlier. As such, it requires that all connection information must be provided in the PROC SQL statement. This example is the same as the CONNECT USING example above, but it uses the CONNECT TO statement with all connection information:

PROC FEDSQL

The syntax of the FedSQL language is based on the ANSI SQL-99 core standard. It provides a data source neutral SQL dialect that is common across all data sources that it supports. This removes the need to submit queries in data source specific SQL dialect and provides the ability to reference data from various data sources in the same SQL statement. Using the libref defined previously, this example does the same task as the PROC SQL example above, but using PROC FEDSQL:

```
proc fedsql;
    create table work.Google as
        select st.stock, st.date, st.close
        from mydb.stocks as st
        where stock='GOOGL';
quit;
```

In the example, a table alias to qualify the column names in the SELECT statement was required because "close" is a SQL-99 reserved word. In the PROC SQL example above, this was not required. To learn more about PROC FEDSQL syntax please see the <u>Base SAS 9.4</u> Procedures Guide.

READING DATA WITHIN SAS VIYA

There are two CAS actions that can be used with SAS data connectors to read data from third-party data sources: LOADTABLE and FEDSQL EXECDIRECT.

loadTable Action

The loadTable action specifies which table to load into the CAS server. Only the CASLIB option, which determines the caslib of the loaded table in CAS, is required for this action. In this example, the loadTable action also defines which data source to load the table from and the Path= option provides the name of the table to load from your data source. Because you provide no other table name, Cars is the name of the resulting CAS table on the CAS server. For example, to load the Cars table in Google BigQuery into CAS, the following loadTable statement would suffice:

```
proc cas;
   session mysess;
   action loadTable / caslib="caslibBigQuery" path="cars";
quit;
```

execDirect Action

The execDirect action in the FEDSQL action set has added the functionality to submit queries to caslibs that point to data sources. This enables implicit SQL pass-through for queries that are submitted through both PROC FEDSQL using a CAS session reference and the CAS procedure that executes the execDirect action. Implicit SQL pass-through is the process of translating a FEDSQL query into the equivalent data source specific SQL dialect so that it can be executed completely in the data source. The advantage to using this method of loading data over the loadTable action is that it improves query response time and helps limit the amount of data that is transferred to the CAS server.

When a FEDSQL query references data in only a single data source, an attempt is made to pass the full query down to the data source for processing. If the query fails to run on the targeted data source, then implicit SQL pass-through fails and the full table is loaded into CAS. The query is then run on the CAS server against the fully loaded table. In summation, the following is required for implicit SQL pass-through:

- All tables in the FEDSQL query must exist in a single data source.
- All tables in the FEDSQL query must reside in a data source. None of the tables referenced in the query can have been previously loaded into the CAS server prior to the query executing.
- The target data source must be able to run the SQL query.

To enable implicit SQL pass-through, references to the tables in the query that is submitted in the execDirect action must be fully qualified, including the caslib in the table reference in the FROM clause. This takes the form of caslib tablename. That is, the query uses the CASLIB defined previously and the Class table, CaslibBigQuery.Class. Also, for data sources that have case sensitive table and column names, such as Google BigQuery, it is required to put each object in double quotations, i.e. "caslibBigQuery"."class". In the following example, we use the execDirect action to submit a query that returns all 2016 Fords with six cylinders from a Cars table stored in Google BigQuery using the CASLIB defined previously:

```
proc cas;
    session mysess;
    action fedsql.execdirect
        query="select * from ""caslibBigQuery"".""cars""
        where make='Ford' and Cylinders=6";
quit;
```

If implicit SQL pass-through is successful, a note is printed in the log.

```
73 proc cas;
74 session mysess;
75 action fedsql.execdirect
76 query="select * from ""caslibBigQuery"".""cars"" where
make='Ford' and Cylinders=6";
77 run;
NOTE: Active Session now mysess.
NOTE: Added action set 'fedsql'.
NOTE: The SQL statement was fully offloaded to the underlying data source
via full pass-through
```

Output 1. Example of the note in the SAS log from successful implicit SQL pass-through execution.

As mentioned earlier, the second method for submitting queries to the data source using implicit pass-through in the CAS server is through PROC FEDSQL by setting the SESSREF= option to your CAS session:

```
proc fedsql sessref=mysess;
    create table public.fords as select * from "caslibBigQuery"."cars"
        where make='Ford' and Cylinders=6;
quit;
```

In the example above, notice the table reference is again fully qualified with the caslib and table name. Also, this example demonstrates how to save the results of the query into a new table called Fords loaded into the Public caslib.

WRITING DATA

WRITING DATA WITHIN SAS 9

Using SAS/ACCESS Interface to Google BigQuery to write data is similar to any other SAS/ACCESS product.

DATA Step

The example of using the DATA step above to read data created a data set in the Work libref. To write the same data set back to Google BigQuery, you only to switch the source and destination. In this example, the DATA step creates a new table called Google in Google BigQuery by copying the data from the data set into it:

```
data mydb.Google;
    set work.Google;
run;
```

PROC SOL and PROC FEDSOL

Just like in the DATA step example, where all that is necessary to write data is to switch the source and the destination, similar is true for PROC SQL and PROC FEDSQL. Using the

examples from before but in reverse, here is an example of how to write a data set into Google BigQuery:

```
proc sql;
    create table mydb.Google as
        select *
        from work.Google;
quit;

proc fedsql;
    create table mydb.Google as
        select *
        from work.Google;
quit;
```

WRITING DATA WITHIN SAS VIYA

With SAS Viya, the only method to write data from the CAS server to a data source is through the save action. Only the CASLIB= and NAME= options are required for the save action. The CASLIB= option points to where the data is loaded in CAS, and the NAME= option specifies the name of the table. The following example uses the save action to write the Cars table that was loaded into CAS from Google BigQuery in the loadTable action example above back to Google BigQuery:

```
proc cas;
   session mysess;
   action save / caslib="caslibBigQuery" name="cars";
quit;
```

During the execution of the save action, a CREATE TABLE SQL statement is run against Google BigQuery. The save action always attempts to create a new table. If a table with the same name as the one that is being written already exists in Google BigQuery, the action fails unless the REPLACE= option is used and set to TRUE.

If you want to write only a portion of the table to the database, then specify the WHERE= and VARS= options in the TABLE= option. Here is an example that uses these options to limit the number of rows written. Only rows where the year is "2016" and only the Make column is written:

```
proc cas;
   session mysess;
   action save / caslib="caslibBigQuery" name="cars2016"
      table={caslib="public" name="usedcars"
            vars={"make"}
            where="year = 2016"};
quit;
```

This example also demonstrates how to reference data that is loaded into another caslib with a different table name when writing into another caslib.

CONCLUSION

The SAS/ACCESS Interface to Google BigQuery enables you to access Google BigQuery data in several ways. The purpose of this paper is to merely introduce you to this new product through the various examples above and hopefully get you interested in learning more. If it was successful in doing so, please be sure to read the SAS/ACCESS documentation. It is

the best source for information on how to use this product and can help you in getting the most from your investment in SAS.

ACKNOWLEDGMENTS

The author extends his heartfelt thanks and gratitude to the following individuals:

Salman Maher, SAS Institute Inc., Cary, NC

Brad Thompson, SAS Institute Inc., Cary, NC

Chris Dehart, SAS Institute Inc., Cary, NC

Heike Czichy, SAS Institute Inc., Cary, NC

Mason Morris, SAS Institute Inc., Cary, NC

RECOMMENDED READING

- SAS Institute Inc. 2019. "SAS/ACCESS Interface to Google BigQuery." SAS Cloud Analytic Services:
 User's Guide. Available at
 https://go.documentation.sas.com/?cdcld=pgmsascdc&cdcVersion=9.4_3.5&docsetId=acreldb&docsetTarget=n0bm9s8t6t6rr2n1qgkamlgtb0kk.htm&locale=en
- SAS Institute Inc. 2019. "Working with SAS Data Connectors." SAS Cloud Analytic Services: User's Guide. Available at https://documentation.sas.com/?cdcld=pgmsascdc&cdcVersion=9.4-3.5&docsetId=casref&docsetTarget=p1ez56agp5uvukn1f96jscujvplm.htm
- SAS Institute Inc. 2019. "Quick Reference for Data Connector Syntax." SAS Cloud Analytic Services:
 User's Guide. Available at
 https://documentation.sas.com/?cdcld=pgmsascdc&cdcVersion=9.4_3.5&docsetId=casref&docsetTarget=n0wxsz0rzamws5n1ldbepubijenk.htm

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Joel Odom 100 SAS Campus Drive Cary, NC 27513 SAS Institute Inc. Joel.Odom@sas.com http://www.sas.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.